AmirHosein Sadeghimanesh
2020 February, modified 2021 July

This woeksheet contains the computation for finding the CAD representation of the multistationarity region of the bistable autoregulatory motif introduced in Figure 3a of the paper, which is depicted in Figure 4a.

Consider the following parametric system of equations which consists of the steady state equations of three species and a conservation law of the model.

```
> F := [-k[5]*x[1]*x[3] + k[6]*x[4], -2*k[3]*x[2]^2 + k[1]*x[1] - k
  [2]*x[2] + 2*k[4]*x[3] + k[7]*x[4], k[3]*x[2]^2 - k[5]*x[1]*x[3]
  - k[4]*x[3] + k[6]*x[4], x[1] + x[4] - k[8]];
  <seq(polynomial, polynomial in F)>;
```

$$F := \left[-k_5 x_1 x_3 + k_6 x_4, -2 k_3 x_2^2 + k_1 x_1 - k_2 x_2 + 2 k_4 x_3 + k_7 x_4, k_3 x_2^2 - k_5 x_1 x_3 - k_4 x_3 + k_6 x_4, x_1 + x_4 - k_8\right]$$

$$\begin{bmatrix} -k_5 x_1 x_3 + k_6 x_4 \\ -2 k_3 x_2^2 + k_1 x_1 - k_2 x_2 + 2 k_4 x_3 + k_7 x_4 \\ k_3 x_2^2 - k_5 x_1 x_3 - k_4 x_3 + k_6 x_4 \\ x_1 + x_4 - k_8 \end{bmatrix} \tag{1}$$

Next we fix the values of all parameters other than k[3] and k[8].

```
> kk := [2.81, 1, 0.001, 0.98, 2.76, 1.55, 46.9, 1]: # A choice of
  parameter values coming from a reference in the paper.
  eq := [ seq( eval( polynomial = 0, [ seq( k[i] = kk[i], i in [1,
  2, 4, 5, 6, 7] ) ] ), polynomial in F ) ]:
  <seq(polynomial, polynomial in eq)>;
```

$$\begin{bmatrix} -2.7600000000 x_1 x_3 + 1.5500000000 x_4 = 0 \\ -2 k_3 x_2^2 + 2.8100000000 x_1 - x_2 + 1.9600000000 x_3 + 46.9000000000 x_4 = 0 \\ k_3 x_2^2 - 2.7600000000 x_1 x_3 - 0.9800000000 x_3 + 1.5500000000 x_4 = 0 \\ x_1 + x_4 - k_8 = 0 \end{bmatrix} \tag{2}$$

Preparing the input list of equations and inequalities for the RootFinding:-Parametric:-CellDecomposition command of Maple.

```
> eqCAD := [-276/100*x[1]*x[3] + 155/100*x[4] = 0, -2*k[3]*x[2]^2 +
  281/100*x[1] - x[2] + 196/100*x[3] + 469/10*x[4] = 0, -276/100*x
```

```
[1]*x[3] + k[3]*x[2]^2 - 98/100*x[3] + 155/100*x[4] = 0, x[1] + x
[4] - k[8] = 0, 0 < x[1], 0 < x[2], 0 < x[3], 0 < x[4], 0 < k[3],
0 < k[8]]: # Equations and inequalities. Note that RootFinding:-
Parametric package does not accept float numbers, it only accepts
rational coefficients.
```

```
> with(RootFinding:-Parametric): # Loading the package.
st := time[real]():
C := CellDecomposition(eqCAD, [x[1], x[2], x[3], x[4]], [k[3], k
[8]]): # Asking Maple to compute the open cells of the CAD with
respect to the discriminant variety of the system in eqCAD.
NumberOfSolutions(C); # Asking Maple to tell us how many
solutions exist for the points in each cell.
time[real]() - st; # the computation time
```

$$[[1, 1], [2, 3], [3, 1]]$$

$$0.1160000000 \tag{3}$$

Therefore there are 2 open cells with 1 solution, and 1 open cell with three solutions.

Plotting the CAD representation of the multistationarity region. Yellow is used for the multistationarity region and cyan for the opposite.

```
> CellPlot(C, 2, color = yellow, background = ColorTools:-Color(
[0.57,0.88,1]), view = [0.0005 .. 0.0010, 0 .. 2], 'labels' =
[k__3, k__8], 'labelfont' = ["TimesNewRoman", 18], size=[500,460]
); # Plotting the only open cell that has three solutions. This
is the multistationarity region that we are looking for.
```



End of the file.